G13CBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G13CBF calculates the smoothed sample spectrum of a univariate time series using spectral smoothing by the trapezium frequency (Daniell) window.

2 Specification

```
      SUBROUTINE G13CBF(NX, MTX, PX, MW, PW, L, KC, LG, XG, NG, STATS,
     1                  IFAIL)
      INTEGER           NX, MTX, MW, L, KC, LG, NG, IFAIL
      real              PX, PW, XG(KC), STATS(4)
```

3 Description

The supplied time series may be mean or trend corrected (by least-squares), and tapered, the tapering factors being those of the split cosine bell:

$$\begin{array}{ll} \frac{1}{2}\left(1-\cos\left(\frac{\pi(t-\frac{1}{2})}{T}\right)\right), & 1 \le t \le T \\ \frac{1}{2}\left(1-\cos\left(\frac{\pi(n-t+\frac{1}{2})}{T}\right)\right), & n+1-T \le t \le n \\ 1, & \text{otherwise} \end{array}$$

where $T = \left[\frac{np}{2}\right]$ and p is the tapering proportion.

The unsmoothed sample spectrum

$$f^*(\omega) = \frac{1}{2\pi}\left|\sum_{t=1}^{n} x_t \exp(i\omega t)\right|^2$$

is then calculated for frequency values

$$\omega_k = \frac{2\pi k}{K}\ ,\ k = 0,1,\ldots,[K/2]$$

where [ ] denotes the integer part.

The smoothed spectrum is returned as a subset of these frequencies for which k is a multiple of a chosen value r, i.e.,

$$\omega_{rl} = \nu_l = \frac{2\pi l}{L},\ \ l = 0,1,\ldots,[L/2]$$

where $K = r \times L$. The user will normally fix L first, then choose r so that K is sufficiently large to provide an adequate representation for the unsmoothed spectrum, i.e., $K \ge 2 \times n$. It is possible to take $L = K$, i.e., $r = 1$.

The smoothing is defined by a trapezium window whose shape is supplied by the function

$$\begin{array}{ll} W(\alpha) = 1, & |\alpha| \le p \\ W(\alpha) = \frac{1-|\alpha|}{1-p}, & p < |\alpha| \le 1 \end{array}$$

the proportion p being supplied by the user.

The width of the window is fixed as $2\pi/M$ by the user supplying M. A set of averaging weights are constructed:

$$W_k = g \times W\left(\frac{\omega_k M}{\pi}\right)\ ,\ 0 \le \omega_k \le \frac{\pi}{M}$$

where g is a normalising constant, and the smoothed spectrum obtained is

$$\hat{f}(\nu_l) = \sum_{|\omega_k| < \frac{\pi}{M}} W_k f^*(\nu_l + \omega_k).$$

If no smoothing is required M should be set to n, in which case the values returned are $\hat{f}(\nu_l) = f^*(\nu_l)$. Otherwise, in order that the smoothing approximates well to an integration, it is essential that $K \gg M$, and preferable, but not essential, that K be a multiple of M. A choice of $L > M$ would normally be required to supply an adequate description of the smoothed spectrum. Typical choices of $L \simeq n$ and $K \simeq 4n$ should be adequate for usual smoothing situations when $M < n/5$.

The sampling distribution of $\hat{f}(\omega)$ is approximately that of a scaled χ^2_d variate, whose degrees of freedom d is provided by the routine, together with multiplying limits mu, ml from which approximate 95% confidence intervals for the true spectrum $f(\omega)$ may be constructed as $[ml \times \hat{f}(\omega),\ mu \times \hat{f}(\omega)]$. Alternatively, $\log \hat{f}(\omega)$ may be returned, with additive limits.

The bandwidth b of the corresponding smoothing window in the frequency domain is also provided. Spectrum estimates separated by (angular) frequencies much greater than b may be assumed to be independent.

4 References

[1] Jenkins G M and Watts D G (1968) *Spectral Analysis and its Applications* Holden–Day

[2] Bloomfield P (1976) *Fourier Analysis of Time Series: An Introduction* Wiley

5 Parameters

1: NX — INTEGER *Input*

On entry: the length of the time series, n.

Constraint: NX ≥ 1.

2: MTX — INTEGER *Input*

On entry: whether the data are to be initially mean or trend corrected.

MTX = 0 for no correction,

MTX = 1 for mean correction,

MTX = 2 for trend correction.

Constraint: $0 \leq$ MTX ≤ 2.

3: PX — ***real*** *Input*

On entry: the proportion of the data (totalled over both ends) to be initially tapered by the split cosine bell taper. (A value of 0.0 implies no tapering).

Constraint: $0.0 \leq$ PX ≤ 1.0.

4: MW — INTEGER *Input*

On entry: the value of M which determines the frequency width of the smoothing window as $2\pi/M$. A value of n implies no smoothing is to be carried out.

Constraint: $1 \leq$ MW $\leq$ NX.

5: PW — ***real*** *Input*

On entry: the shape parameter, p, of the trapezium frequency window.

A value of 0.0 gives a triangular window, and a value of 1.0 a rectangular window.

If MW = NX (i.e., no smoothing is carried out), then PW is not used.

Constraint: $0.0 \leq$ PW ≤ 1.0.

6: L — INTEGER *Input*

On entry: the frequency division, L, of smoothed spectral estimates as $2\pi/L$.

Constraints: $L \geq 1$,

L must be a factor of KC (see below).

7: KC — INTEGER *Input*

On entry: the order of the fast Fourier transform (FFT), K, used to calculate the spectral estimates. KC should be a multiple of small primes such as 2^m where m is the smallest integer such that $2^m \geq 2n$, provided $m \leq 20$.

Constraints:

$KC \geq 2 \times NX$,
KC must be a multiple of L. The largest prime factor of KC must not exceed 19, and the total number of prime factors of KC, counting repetitions, must not exceed 20. These two restrictions are imposed by C06EAF which performs the FFT.

8: LG — INTEGER *Input*

On entry: indicates whether unlogged or logged spectral estimates and confidence limits are required.

$LG = 0$ for unlogged.

$LG \neq 0$ for logged.

9: XG(KC) — ***real*** array *Input/Output*

On entry: the n data points.

On exit: contains the NG spectral estimates $\hat{f}(\omega_i)$, for $i = 0, 1, \ldots, [L/2]$, in XG(1) to XG(NG) (logged if $LG \neq 0$). The elements XG(i), for $i = NG + 1, \ldots, KC$ contain 0.0.

10: NG — INTEGER *Output*

On exit: the number of spectral estimates, $[L/2] + 1$, in XG.

11: STATS(4) — ***real*** array *Output*

On exit: four associated statistics. These are the degrees of freedom in STATS(1), the lower and upper 95% confidence limit factors in STATS(2) and STATS(3) respectively (logged if $LG \neq 0$), and the bandwidth in STATS(4).

12: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).

For this routine, because the values of output parameters may be useful even if IFAIL $\neq 0$ on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit**. To suppress the output of an error message when soft failure occurs, set IFAIL to 1.

6 Error Indicators and Warnings

Errors or warnings specified by the routine:

IFAIL = 1

On entry, NX < 1,
or MTX < 0,
or MTX > 2,
or PX < 0.0,
or PX > 1.0,
or MW < 1,
or MW $>$ NX,
or PW < 0.0 and MW $\neq$ NX,
or PW > 1.0 and MW $\neq$ NX,
or L < 1.

IFAIL = 2

On entry, KC $< 2 \times$ NX,
or KC is not a multiple of L,
or KC has a prime factor exceeding 19,
or KC has more than 20 prime factors, counting repetitions.

IFAIL = 3

This indicates that a serious error has occurred. Check all array subscripts and subroutine parameter lists in calls to G13CBF. Seek expert help.

IFAIL = 4

One or more spectral estimates are negative. Unlogged spectral estimates are returned in XG, and the degrees of freedom, unlogged confidence limit factors and bandwidth in STATS.

IFAIL = 5

The calculation of confidence limit factors has failed. This error will not normally occur. Spectral estimates (logged if requested) are returned in XG, and degrees of freedom and bandwidth in STATS.

7 Accuracy

The FFT is a numerically stable process, and any errors introduced during the computation will normally be insignificant compared with uncertainty in the data.

8 Further Comments

G13CBF carries out a FFT of length KC to calculate the sample spectrum. The time taken by the routine for this is approximately proportional to KC $\times$ log(KC) (but see Section 8 of the document for C06EAF for further details).

9 Example

The example program reads a time series of length 131. It selects the mean correction option, a tapering proportion of 0.2, the option of no smoothing and a frequency division for logged spectral estimates of $2\pi/100$. It then calls G13CBF to calculate the univariate spectrum and prints the logged spectrum together with 95% confidence limits. The program then selects a smoothing window with frequency width $2\pi/30$ and shape parameter 0.5 and recalculates and prints the logged spectrum and 95% confidence limits.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G13CBF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          KCMAX, NXMAX
      PARAMETER        (KCMAX=400,NXMAX=KCMAX/2)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             PW, PX
      INTEGER          I, IFAIL, KC, L, LG, MTX, MW, NG, NX
*     .. Local Arrays ..
      real             STATS(4), XG(KCMAX), XH(NXMAX)
*     .. External Subroutines ..
      EXTERNAL         G13CBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13CBF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX
      IF (NX.GT.0 .AND. NX.LE.NXMAX) THEN
         READ (NIN,*) (XH(I),I=1,NX)
         MTX = 1
         PX = 0.2e0
         MW = NX
         PW = 0.5e0
         KC = 400
         L = 100
         LG = 1
   20    READ (NIN,*,END=60) MW
         IF (MW.GT.0 .AND. MW.LE.NX) THEN
            DO 40 I = 1, NX
               XG(I) = XH(I)
   40       CONTINUE
            IFAIL = 1
*
            CALL G13CBF(NX,MTX,PX,MW,PW,L,KC,LG,XG,NG,STATS,IFAIL)
*
            WRITE (NOUT,*)
            IF (MW.EQ.NX) THEN
               WRITE (NOUT,*) 'No smoothing'
            ELSE
               WRITE (NOUT,99999)
     +           'Frequency width of smoothing window = 1/', MW
            END IF
            WRITE (NOUT,*)
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,99999) 'G13CBF fails. IFAIL =', IFAIL
               WRITE (NOUT,*)
            END IF
            IF (IFAIL.EQ.0 .OR. IFAIL.GE.4) THEN
               WRITE (NOUT,99998) 'Degrees of freedom =', STATS(1),
     +           '      Bandwidth =', STATS(4)
               WRITE (NOUT,*)
               WRITE (NOUT,99997)
```

```
     +          '95 percent confidence limits -     Lower =', STATS(2),
     +          '  Upper =', STATS(3)
               WRITE (NOUT,*)
               WRITE (NOUT,*)
     +    '      Spectrum       Spectrum       Spectrum       Spectrum'
               WRITE (NOUT,*)
     +    '      estimate       estimate       estimate       estimate'
               WRITE (NOUT,99996) (I,XG(I),I=1,NG)
            END IF
            GO TO 20
         END IF
      END IF
   60 STOP
*
99999 FORMAT (1X,A,I3)
99998 FORMAT (1X,A,F4.1,A,F7.4)
99997 FORMAT (1X,A,F7.4,A,F7.4)
99996 FORMAT (1X,I4,F10.4,I5,F10.4,I5,F10.4,I5,F10.4)
      END
```

9.2 Program Data

```
G13CBF Example Program Data
 131
 11.500  9.890  8.728  8.400  8.230  8.365  8.383  8.243
  8.080  8.244  8.490  8.867  9.469  9.786 10.100 10.714
 11.320 11.900 12.390 12.095 11.800 12.400 11.833 12.200
 12.242 11.687 10.883 10.138  8.952  8.443  8.231  8.067
  7.871  7.962  8.217  8.689  8.989  9.450  9.883 10.150
 10.787 11.000 11.133 11.100 11.800 12.250 11.350 11.575
 11.800 11.100 10.300  9.725  9.025  8.048  7.294  7.070
  6.933  7.208  7.617  7.867  8.309  8.640  9.179  9.570
 10.063 10.803 11.547 11.550 11.800 12.200 12.400 12.367
 12.350 12.400 12.270 12.300 11.800 10.794  9.675  8.900
  8.208  8.087  7.763  7.917  8.030  8.212  8.669  9.175
  9.683 10.290 10.400 10.850 11.700 11.900 12.500 12.500
 12.800 12.950 13.050 12.800 12.800 12.800 12.600 11.917
 10.805  9.240  8.777  8.683  8.649  8.547  8.625  8.750
  9.110  9.392  9.787 10.340 10.500 11.233 12.033 12.200
 12.300 12.600 12.800 12.650 12.733 12.700 12.259 11.817
 10.767  9.825  9.150
 131
  30
```

9.3 Program Results

```
G13CBF Example Program Results

No smoothing

Degrees of freedom = 2.0      Bandwidth = 0.0480

95 percent confidence limits -     Lower =-1.3053  Upper = 3.6762

      Spectrum        Spectrum        Spectrum        Spectrum
      estimate        estimate        estimate        estimate
   1   -5.9354    2   -0.1662    3   -0.8250    4   -0.9452
   5    3.2137    6    0.2738    7   -1.0690    8   -1.0401
   9   -1.2388   10   -3.5434   11   -5.2568   12   -3.2450
  13   -2.4294   14   -3.9987   15   -2.9853   16   -4.6631
  17   -4.3317   18   -4.6982   19   -4.6335   20   -3.6732
  21   -5.8411   22   -4.7727   23   -3.9747   24   -4.8351
  25   -5.9979   26   -6.1169   27   -5.5245   28   -4.4774
  29   -5.6331   30   -4.0707   31   -4.6921   32   -5.6515
  33   -9.2919   34   -4.6302   35   -4.1700   36   -4.7829
  37   -6.6058   38   -5.8145   39   -5.2714   40   -5.8736
  41  -10.2188   42   -5.7887   43   -7.0751   44   -7.4055
  45   -8.2774   46   -7.8966   47   -6.4435   48   -5.7844
  49   -5.4690   50   -6.8709   51   -8.7123

Frequency width of smoothing window = 1/ 30

Degrees of freedom = 7.0      Bandwidth = 0.1767

95 percent confidence limits -     Lower =-0.8275  Upper = 1.4213

      Spectrum        Spectrum        Spectrum        Spectrum
      estimate        estimate        estimate        estimate
   1   -0.1776    2   -0.4561    3   -0.1784    4    1.9042
   5    2.1094    6    1.7061    7   -0.7659    8   -1.4734
   9   -1.5939   10   -2.1157   11   -2.9151   12   -2.7055
  13   -2.8200   14   -3.4077   15   -3.8813   16   -3.6607
  17   -4.0601   18   -4.4756   19   -4.2700   20   -4.3092
  21   -4.5711   22   -4.8111   23   -4.5658   24   -4.7285
  25   -5.4386   26   -5.5081   27   -5.2325   28   -5.0262
  29   -4.4539   30   -4.4764   31   -4.9152   32   -5.8492
  33   -5.5872   34   -4.9804   35   -4.8904   36   -5.2666
  37   -5.7643   38   -5.8620   39   -5.5011   40   -5.7129
  41   -6.3894   42   -6.4027   43   -6.1352   44   -6.5766
  45   -7.3676   46   -7.1405   47   -6.1674   48   -5.8600
  49   -6.1036   50   -6.2673   51   -6.4321
```